

Lambdas Are Simple, or How To CPS

Paulette Koronkevich

October 23, 2018

We've been talking about how to convert functions to *continuation passing style*, but what does that even mean? A *continuation* is a way to analyze the *control flow* of your program at any point. The way we represent a continuation in CSCI-C311 is with a *function*. We transform our programs to “push” the computation into the continuation, so that we can track its evaluation. Now, here's a quick introduction on how to CPS (convert into continuation passing style) programs.

1 Change the name

Change the name of the program from `function-name` to `function-name-cps`. Easy enough, right?

2 Add a k to the formal parameters

```
(define function-name (λ (x1 ... xn) ...))
```

changes to

```
(define function-name-cps (λ (x1 ... xn k) ...))
```

Again, easy enough, right? This `k` represents the continuation. Remember that it is a function.

3 Apply k to simple values

Simple values include (but are not limited to) numbers, lists, booleans, strings, symbols, and most importantly - **lambdas are simple**.

```
(define add2 (λ (x) (+ 2 x)))
```

changes to

```
(define add2-cps (λ (x k) (k (+ 2 x))))
```

```
(define if0 (λ (x) (if (zero? x) 'zero 'not-zero)))
```

changes to

```
(define if0-cps (λ (x k) (if (zero? x) (k 'zero) (k 'not-zero))))
```

Notice that `k` is applied to the values *returned* by the function. Applying `k` to the entire `if` statement might work, but `if` isn't a simple value - it's a way to introduce a branch in our program. We are trying to track how the values change throughout the program. Applying the continuation to simple values invokes the current “stack” of computation onto the value.

```
(define cur-mult (λ (n) (λ (m) (* n m))))
```

changes to

```
(define cur-mult-cps (λ (n k) (k (λ (m k) (k (* n m))))))
```

Notice that `k` is added to the formal parameters of the function returned by `cur-mult`. This is because *all functions used and returned by the program must also be in continuation passing style*. `cur-mult` is a function that, when given a number, returns a function. Lambdas are simple, so `k` must be applied onto this simple value `cur-mult` returns. The function returned also takes a number, and then multiplies the two. This function must be CPS'd along with `cur-mult`. Thus, we add `k` to the formal parameters, and apply `k` to the simple value it returns.

4 Convert all “serious” calls to tail calls

Um, what? Let's examine this step closely. First of all, “serious” calls are (1) recursive function calls, (2) calls to other, previously defined functions and (3) calling a function passed as an argument. Let's identify all serious calls in the following function:

```
(define srs-calls
  (lambda (f x)
    (if (zero? x) (f x) (+ 1 (srs-calls f (sub1 x))))))
```

There's an application of some argument called `f` onto the argument `x`, and also a recursive call to `srs-calls`. Thus, the two “serious” calls are highlighted in red:

```
(define srs-calls
  (lambda (f x)
    (if (zero? x) (f x) (+ 1 (srs-calls f (sub1 x))))))
```

Alright, so what is a “tail call”? Unfortunately, it doesn't have to do with any furry tailed friends. A *tail call* is the application of a function where no computation is “waiting” on it. Referring back to our example, `(f x)` has no computation waiting on it, whereas `(srs-calls f (sub1 x))` has the `+ 1` waiting on its result.

4.1 Conversion

Now, how do we do this conversion? First and foremost, recall that *all functions used and returned by the program must also be in continuation passing style*. In order to convert serious calls to tail calls, we defer any computation waiting on it to its continuation. Let's see some simple examples.

```
(define simple (λ (f x) (f x)))
```

Applying the first two steps of CPSing:

```
(define simple-cps (λ (f x k) (f x)))
```

`f` must be in continuation passing style, so it must have a continuation. Recall that the continuation is a function that tracks the control flow of the program. We can start by writing a lambda to start constructing the continuation (function)...

```
(define simple-cps (λ (f x k) (f x (λ (res) ...))))
```

I named the formal parameter of this function `res` to emphasize the point that this continuation associated with this serious call will take as input the *result* of calling `f` on `x`! Once we have this result, we can apply any computation waiting on it, like the `+ 1` that waits on the serious call `(srs-calls f (sub1 x))` in the previous example. In this case, no computation is waiting

on the result of `f` applied to `x`. This result is simple, since we can assume that `f` will terminate with a simple value. And what do we do with simple values? We apply `k`!

```
(define simple-cps (λ (f x k) (f x (λ (res) (k res))))))
```

Finally, notice that `(λ (res) (k res))` is equivalent to just `k`. This is called *eta-reduction*. It's not a necessary step, but makes our code look a lot cleaner in most cases, and I will be applying this reduction to future examples. Finally, we have:

```
(define simple-cps (λ (f x k) (f x k)))
```

Alright, more examples! Let's go back to `srs-calls`, now with the first two steps of the transformation applied:

```
(define srs-calls-cps
  (lambda (f x k)
    (if (zero? x) (f x) (+ 1 (srs-calls f (sub1 x))))))
```

`(f x)` is easy enough to convert to a tail call - it's already one!

```
(define srs-calls-cps
  (lambda (f x k)
    (if (zero? x) (f x k) (+ 1 (srs-calls f (sub1 x))))))
```

`(srs-calls f (sub1 x))` is not a tail call, so we need to “push” the computation that is waiting on the result into the continuation.

```
(define srs-calls-cps
  (lambda (f x k)
    (if (zero? x)
        (f x k)
        (srs-calls f (sub1 x) (λ (res) (+ 1 res))))))
```

Finally, don't forget what we do with simple values! Apply `k`.

```

(define srs-calls-cps
  (lambda (f x k)
    (if (zero? x)
        (f x k)
        (srs-calls f (sub1 x) (λ (res) (k (+ 1 res)))))))

```

A good rule of thumb to check through your program and make sure there is no computation waiting on a serious call. If you ever see something like `(add1 (f x))` or `(cons (f x) '())`, you're not done or have done something wrong. Keep applying this technique, “pushing” this computation into the continuation of the serious call.

Also, functions like `cons`, `append`, `plus`, `minus`, etc are **not serious**. These are operations over simple values, and return simple values.